类 文件搜索器
	@隐藏
	变量 返回集合 : 文本集合
	@隐藏
	变量 Android : 安卓窗口?
	@隐藏
	变量 是否搜索文件夹 : 逻辑型 = 假
	
	
	
	方法 搜索文件(路径 : 文本,搜索名称 : 文本,窗口环境 : 安卓窗口,搜索文件夹 : 逻辑型 = 真)
		返回集合.清空()
		是否搜索文件夹=搜索文件夹
		变量 pk : 文件=路径
		Android=窗口环境
		提交到新线程运行()
		SearchFile(搜索名称,pk,返回集合)
		提交到主线程运行(Android)
		搜索文件结束(返回集合)
		结束提交到主线程()
		结束提交到新线程()
	结束 方法
	
	@隐藏
	方法 SearchFile(Name : 文本 , Path : 文件 , return : 文本集合)
		如果 Path.可读==真 则
			变量 arrayPath : 文本集合 = 文件操作.取子文件集合(Path.取绝对路径())
			 循环(arrayPath->i)
				变量 File : 文件 = i
				
				提交到主线程运行(Android)
				正在搜索文件(File.取绝对路径(),返回集合.长度)
				结束提交到主线程()
				
				如果 File.为文件() 则
					如果 File.取文件名() ? Name 则
						return.添加成员(i)
					结束 如果
				否则
					如果 (File.取文件名() ? Name) 且 是否搜索文件夹==真 则
						return.添加成员(i)
					结束 如果
					SearchFile(Name,File,return)
				结束 如果
			结束 循环
		结束 如果
	结束 方法
	
	
	
	定义事件 正在搜索文件(文件路径 : 文本,文件数目 : 整数)
	
	//在使用列表框时一定要清空列表框显示的东西不然索引会越界
	//可以 列表框.更新项目(0)，简单粗暴
	定义事件 搜索文件结束(文件列表 : 文本集合)
	
结束 类